
竹内寿一郎

　
５．Down and Out Call (B ≤ K)の場合の評価式
　このケースは前々回のNo.9に相当し【3】、バリアー価格 (B)がスポット価格より小さい (こ
れをDownという)とき、ノックアウトで、コールオプション、行使価格 (K)がバリアー価
格 (B)より大きい場合である。たとえば、ここ 5節では次のような取引である。
　　　原資産　　　　　ドル円為替レート　
　　　満期　　　　　　 6ヶ月後
　　　スポット価格　　 120円
　　　バリアー価格　　 90円
　　　行使価格　　　　 100円
　　　オプション　　　ノックアウトコールオプション
現在の円の対ドル為替レートが 120円のときの、行使価格 100円のコールオプションではあ
るが、オプション期間の 6ヶ月以内に為替レートが、1度でも 90円をつけたときこのオプショ
ンは消滅してしまう、ということである。そういうときの評価式を求めることになる。

図 1．Downのとき、満期に価格が行使価格以上になる 2つの経路
　　 (行使価格 100円、バリアー価格 90円、スポット価格 120円）

　
　バリアー価格を 90円として、次の 2つの経路を考える

[1] 満期Tに価格が行使価格 100円以上になる場合 (ここでは過去の経路は一切問わない)、

[2] オプション期間中に一度はバリアー価格 90円以下となり、その後満期 Tでは価格が
100円以上となる。

　オプション期間中に一度もバリアー価格に達せず (バリアー価格以下にならず)、オプショ
ン満期Tに価格S(T )が、行使価格Kより大きくなっているとき、キャッシュフローS(T )−K

が得られる評価額を C(これが求める評価額)、経路 [1]での価格を CI、経路 [2]での価格を
CII で表すとすると、
　　C = CI − CII

と書くことができる。
　経路 [2]は、経路 [1]が過去の経路は一切問わないので [1]の部分であり、バリアーに到達
すること無く価格 S(T)が行使価格K以上になる確率は、[1]の確率から [2]の確率を引けば
よい。すなわち、このときの評価額は [1]のときの評価額 CI から [2]のときの評価額 CII を
引けば良いことになる。
　そこでまず、満期に価格が行使価格以上になるとき、キャッシュフロー S(T ) − Kを得る
オプションの評価額CI を計算してみよう。
　　C1：満期 T で価格が行使価格以上のときの価格
とするとCI は
　　CI = C1

で表される。
　 [1]の確率は単に満期に行使価格以上と言う条件なので、これはプレインバニラオプショ
ンの確率に一致する（補遺または文献【2】を参照のこと）。従って、
(1)　　C1 = S(t)Φ(d1 + σ

√
T − t) − Ke−r(T−t)Φ(d1)

ここで、

(2)　　 d1 =
lnS(t)

K
+ (r − σ2

2
)(T − t)

σ
√

T − t
である。
　 CII の計算は前回におけるmを使って表現することができる【4】。いま、S(・)がBより
ずっと小さい Z(t)から出発して満期 T で (B ≤ K ≤)S(T )に到達する経路に注目し、この
経路上で初めてBに到達する時点を t0とする。つまり、出発点 tでは、

(3)　　Z(t) =
B2

S(t)
： BはZ(t)と S(t)の幾何平均

であるとする。

図 2．Bはスポット価格 S(t)とZ(t)の幾何平均である
ただし、図の左が大きく、右が小さく、通常の大小のスケールを逆にしている

　
　ブラウン運動の定義から系列は過去の時点や、経路に無関係なので t0でBに到達した後
の経路は S(t)を出発点としても、Z(t)を出発点としても t0以降は全く同じ性質を持って動
く。
　そこでZ(t)を出発点として満期でキャッシュフロー S(T )−Kを得る経路に対する評価額
をC ′

IIとすると、S(t)を出発点とする経路に対する評価額CIIとの違いは、出発点からBに

初めて到達する確率mの違いだけに起因するものである。従って、その違いだけを考慮すれ
ばC ′

II からCII を求めることが出来る。そこでまずC ′
II を求めてみる。

　Z(t)を出発してBを超えて満期で (B ≤ K ≤)S(T )となるオプション価格の評価額は、途
中のバリアーには関係しないので (1)と同じようにプレインバニラオプションの期待値で計
算できる。
(4)　　C ′

II = C3

(5)　　C3 = E∞
K {max(0, S(T ) − K)|Z(t)}（ここでの期待値の積分範囲Kは無くてもよい）

　　　　　　= Z(t)Φ(d3 + σ
√

T − t) − e−γ(T−t)KΦ(d3)

　　　　　　=
B2

S(t)
Φ(d3 + σ

√
T − t) − e−γ(T−t)KΦ(d3)

(6)　　 d3 =
lnZ(t)

K
+ (r − σ2

2
)(T − t)

σ
√

T − t
, ここで、Z(t)を S(t)で表して、

(7)　　 d3 =
ln B2

KS(t)
+ (r − σ2

2
)(T − t)

σ
√

T − t
　以上をまとめると、
(8)　　C = CI − CII = CI − mC ′

II = C1 − mC3

(9)　　m =

(
B

S(t)

) 2r
σ2 −1

を得る。
　
　
６．JによるDown and Out Call (B ≤ K)の場合の評価式
　 Jによる関数で、バリアーオプションの評価式をつくり数値を求める。一方、シミュレー
ションによってブラウン運動の系列を発生させ、バリアーに到達したらプレインバニラオプ
ションを消滅させることで、バリアーオプションの価格の期待値を計算して理論値と比較す
ることにする。
　まず恒例により、正規分布に関する J の関数を定義する。ただし、バージョン J6に準拠
しているので、J5以前のバージョンでは引数 x、yを x.、y.に変えなければならない。

NB. Normal Distribution

stnormal=:(%:@o.@2:)(%~)^@-@-:@*:

NP=:3 : 0

(stnormal y)*y%(-‘%‘+‘%)/,(>:+:k),.(*:y)*>:k=.i.28

)

NQ=:3 : 0

(stnormal y)*%‘+/1,,y ,.>:i.28

)

Ndist=:3 : 0

if. 3.3<z=:|y do. q=:NQ z else. q=:0.5-NP z end.

if. 0<:y do. q=:1-q end.

)

NB. Yamanouti’s Formula

Ninv_y=:3 : 0

z=.-^.4*y*(1-y)

x=.%:z*(2.0611786-5.7262204%(z+11.640595))

if. y>0.5 do. x=.-x end.

)

NB. Normal Random Numbers

NB. Rndm_Norm Size Mu Sigma

Rndm_Norm=:3 : 0

’Num Mean Sigma’=.y

Mean+Sigma*Ninv_y"0 (?Num#10000000)%10000000

NB.{.Ninv_bm"1 z=:(?(Num,2)$10000000)%10000000

)

●以下がバリアーオプション No.9の価格式である。
NB. =========================================================

NB. Barrier Option Pricing Model No.09

NB. J.Takeuchi Jan. 2009

NB. Usage: Barrier09 data

NB. data is list ( 6 menbers)

NB. SpotPrice ExePrice BarrierPrice Term(Month) Sigma FreeRate

NB. ex. data=. 120 100 90 6 30 5

NB. =========================================================

Barrier09=:3 : 0

’Spot Kosi Barrier Term Sigma Rate’=. y

Ter=:Term%12[Sig=:Sigma%100[Rat=:Rate%100

m=.(Barrier%Spot)^((2*Rat%Sig^2)-1)

er=.^-Rat*Ter

d1=.ddd Spot,Kosi

d3=.ddd (Barrier^2),Kosi*Spot

C1=.(Spot*Ndist(d1+Sig*%:Ter))-Kosi*er*Ndist(d1)

C3=.(((Barrier^2)%Spot)*Ndist(d3+Sig*%:Ter))-Kosi*er*Ndist(d3)

C=.C1-m*C3

)

NB.============================================================

NB.ddd Numerator(Bunsi),Denominator(Bunbo)

NB.Usage : ddd Spot,Barrier

NB.============================================================

ddd=:3 : 0

’Nume Deno’=.y

((^.Nume%Deno)+(Rat--:*:Sig)*Ter)%Sig*%:Ter

)

NB.============================================================

NB.Simulation for Barrier Option Type 9

NB.Numbers Barrier_09 Spot Kousi Barrier Term Volatility Rate

NB.============================================================

Barrier_09=:4 : 0

’S0 K B T Vol r’=.y

i=.0[Smin=.S=.MM#S0[MM=.x

label_L1.

if. T<i=.>:i do. goto_owari. end.

z=.Rndm_Norm MM,0,1

S=.S+(S*(r%100)*(1%12))+S*(Vol%100)*(%:1%12)*z

Smin=.S<.Smin

NB. print S

goto_L1.

label_owari.

w=.(S-K)*(p=:Smin>B)

NB. w=.(S-K)

C=.(^-(r%100)*(T%12))*MM%~+/(0<w)#w

)

NB. ==========================================

NB. Plain Vannila Option Model

bs =: 3 : 0

’a b c d e’=. y

t=. c % 12

u=. (^. a % b) + t* (e1=.e % 100) - -:(bor=.d % 100) ^2

p2=. u % (bor * %: t)

p1=. p2 + bor * %: t

n1=. Ndist p1

n2=. Ndist p2

bs=. (a * n1 ) - b *n2 *( ^ (-e1) * t)

bs

)

NB. ===========================================

●まず、今回の例題から、スポット価格 120円、
行使価格 100円、バリアー価格 90円、オプション期間 6ヶ月、
リスクフリー金利 5%、ボラティリティ30%では、

Barrier09 120 100 90 6 30 5

24.1793

– 5 –



JAPLA 研究会　 2009/01/24

●バリアーをあげてゆくと、
Barrier09 120 100 85 6 30 5

24.3964

Barrier09 120 100 80 6 30 5

24.4485

Barrier09 120 100 75 6 30 5

24.457

Barrier09 120 100 60 6 30 5

24.458

ほとんどバリアーに関係がなくなる。
ちなみにブラック・ショールズ（プレインバニラオプション）では

bs 120 100 6 30 5

24.458

である。

●期間を長くすると、バリアーの影響が大になる。
Barrier09 120 100 90 12 30 5

27.4263

Barrier09 120 100 85 12 30 5

28.28

Barrier09 120 100 80 12 30 5

28.6727

Barrier09 120 100 75 12 30 5

28.8225

Barrier09 120 100 60 12 30 5

28.8802

bs 120 100 12 30 5

28.8804

Barrier09 120 100 90 24 30 5

31.5252

Barrier09 120 100 85 24 30 5

33.4582

Barrier09 120 100 80 24 30 5

34.7122

Barrier09 120 100 75 24 30 5

35.454

Barrier09 120 100 60 24 30 5

36.0985

bs 120 100 24 30 5

36.1277

●シミュレーションによる検討。
まず、今回の例題から。
スポット価格 120円、行使価格 100円、バリアー価格 90円
、オプション期間 6ヶ月、リスクフリー金利 5%、ボラティリテ
ィ30%では

100000 Barrier_09 120 100 90 6 30 5

24.4125

100000 Barrier_09 120 100 90 6 30 5

24.4883

100000 Barrier_09 120 100 90 6 30 5

24.4883

やや高めに出る。計算値は 24.1793である。
バリアーを低くしてゆくと、

100000 Barrier_09 120 100 85 6 30 5

24.5357

100000 Barrier_09 120 100 80 6 30 5

24.4538

100000 Barrier_09 120 100 75 6 30 5

24.442

100000 Barrier_09 120 100 70 6 30 5

24.5946

100000 Barrier_09 120 100 60 6 30 5

24.5007

100000 Barrier_09 120 100 0 6 30 5

24.6498

100000 Barrier_09 120 100 0 6 30 5

24.5659

プレインバニラ（ブラック・ショールズ）では、バリアーをゼロとして
Barrier09 120 100 0 6 30 5

24.458

【補遺】

　これまでのように価格 S(T )が幾何ブラウン運動をすると仮定すると、
(1)　　 dS(T ) = rS(T )dt + σS(T )dz

なる確率微分方程式を満たす。ここで、rは公平なる利率（無リスク世界における金利）、σ

は S(T )のボラティリティ、dzは標準ブラウン運動の増分を表している。
　そこで始めの時点を t、満期を T とし、伊藤の定理、さらにブラウン運動の分散が σ2×時
間であることを考慮すると【2】、

(2)　　
{lnS(T )/S(t)} − (r − σ2/2(T − t)

σ
√

T − t
∼ 標準正規分布

であることから、満期 T においてキャッシュフローS(T )−Kを得るオプションの評価額は、

(3)　　C1 = E{Max(0, S(T ) − K)} =
∫ ∞

K
(S(T ) − K)f(S(T ))dS(T )

ここで、f(S(T ))dS(T )は S(T )の確率分布で (2)の変換により標準正規分布にしがう。
この式はプレインバニラコールオプションと全く同じなので、
(4)　　C1 = S(t)Φ(d1 + σ

√
T − t) − Ke−r(T−t)Φ(d1)

ここで、

(5)　　 d1 =
lnS(t)

K
+ (r − σ2

2
)(T − t)

σ
√

T − t
である。

